TARGET = qastrologer
TEMPLATE = app

HEADERS += qastrologer.h \
    ihttpclient.h \
    qhttpclient_qhttp.h \
    qconstants.h \
    qzodiac.h \
    qhtmlparser.h \
    qproxydialog.h \
    mainwindow.h

SOURCES += main.cpp \
    qhttpclient_qhttp.cpp \
    qconstants.cpp \
    qzodiac.cpp \
    qhtmlparser.cpp \
    qproxydialog.cpp \
    mainwindow.cpp

RESOURCES += resource.qrc

FORMS += mainwindow.ui proxydialog.ui

win32 {
    RC_FILE = resource.rc
    INCLUDEPATH = "./libcurl/include"
    CONFIG(debug, debug|release) {
        LIBS += "./libcurl/lib/vs2005/libcurld_imp.lib"
    } else {
        LIBS += "./libcurl/lib/vs2005/libcurl_imp.lib"
    }
}
linux-g++ {
    LIBS += -lcurl
}
unix {
    isEmpty(PREFIX):PREFIX = /usr
    BINDIR = $$PREFIX/bin
    DATADIR = $$PREFIX/share
    target.path = $$BINDIR
    icon.path = $$DATADIR/icons/hicolor/scalable/apps
    icon.files += qastrologer.svg
    desktop.path = $$DATADIR/applications
    desktop.files += qastrologer.desktop
    INSTALLS += target icon desktop
}
symbian {
    ICON = qastrologer.svg
    TARGET.CAPABILITY = LocalServices NetworkServices ReadUserData UserEnvironment WriteUserData
    # from 4.7.2
    # http://doc.qt.nokia.com/4.7-snapshot/qmake-variable-reference.html#deployment
    DEPLOYMENT.display_name = QAstrologer
    DEPLOYMENT.installer_header = "$${LITERAL_HASH}{\"QAstrologer Installer\"},(0xA000D7CE),2,0,0"
} else {
    HEADERS += chttpclient.h
    SOURCES += chttpclient.cpp
}

QT += network xml

CONFIG += debug_and_release
